Uge05 SD E20, opgave 5 A

Løs ligningssystemet:

$$\left.\begin{array}{l} a\cdot x_1 + x_2 + x_3 = 1 \\ x_1 + a\cdot x_2 + x_3 = 1 \\ x_1 + x_2 + a\cdot x_3 = 1 \end{array}\right\}$$

Opgaven er helt central.
Den er svær og forudsætter både teori fra matematikken og erfaring med Maple.

VIGTIGT: Når der indgår en parameter (her a) i ligningssystemet, skal man være meget omhyggelig ved løsningen.
Maples rutiner "LinearSolve" og "ReducedRowEchelonForm" tager ikke hensyn til, at opgaven kan få et andet udfald i visse specieltilfælde!
Det er overladt til opgaveløseren.

> *restart*

> *with*(*LinearAlgebra*) :

Koefficientmatricen opskrives:

> $A := \begin{bmatrix} a & 1 & 1 \\ 1 & a & 1 \\ 1 & 1 & a \end{bmatrix}$

$$A := \begin{bmatrix} a & 1 & 1 \\ 1 & a & 1 \\ 1 & 1 & a \end{bmatrix} \tag{1}$$

Højresiden opskrives på matrixform:

> $b := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$

$$b := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \tag{2}$$

Ligningssystemets totalmatrice opskrives:

> $T := \langle A|b \rangle$

$$T := \begin{bmatrix} a & 1 & 1 & 1 \\ 1 & a & 1 & 1 \\ 1 & 1 & a & 1 \end{bmatrix} \tag{3}$$

Metode 1: Løsning ved brug af *RowOperation*

Først laves 3 rækkeoperationer, som er gyldige for alle a :

> *T1* := *RowOperation*(*T*, [1, 3]);
> *T2* := *RowOperation*(*T1*, [2, 1], −1);
> *T3* := *RowOperation*(*T2*, [3, 1], −*a*)

$$T1 := \begin{bmatrix} 1 & 1 & a & 1 \\ 1 & a & 1 & 1 \\ a & 1 & 1 & 1 \end{bmatrix}$$

$$T2 := \begin{bmatrix} 1 & 1 & a & 1 \\ 0 & a-1 & 1-a & 0 \\ a & 1 & 1 & 1 \end{bmatrix}$$

$$T3 := \begin{bmatrix} 1 & 1 & a & 1 \\ 0 & a-1 & 1-a & 0 \\ 0 & 1-a & -a^2+1 & 1-a \end{bmatrix} \tag{1.1}$$

Nu er 1. søjle gjort færdig. Et initial-et-tal i 1. række og 0'er nedenunder.

Da vi nu vil dividere med $a-1$, må vi forudsætte at dette tal ikke er 0!

Antag: $a \neq 1$:

> $T4 := RowOperation\left(T3, 2, \frac{1}{a-1}\right);$
> $T4 := simplify(T4);$
> $T5 := RowOperation(T4, [1, 2], -1);$
> $T6 := RowOperation(T5, [3, 2], a-1)$

$$T4 := \begin{bmatrix} 1 & 1 & a & 1 \\ 0 & 1 & \frac{1-a}{a-1} & 0 \\ 0 & 1-a & -a^2+1 & 1-a \end{bmatrix}$$

$$T4 := \begin{bmatrix} 1 & 1 & a & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 1-a & -a^2+1 & 1-a \end{bmatrix}$$

$$T5 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 1-a & -a^2+1 & 1-a \end{bmatrix}$$

$$T6 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & -a^2-a+2 & 1-a \end{bmatrix} \tag{1.2}$$

Nu er 2. søjle gjort færdig. Et initial-et-tal i 2. række og 0 ovenover og nedenunder.

Matrix-elementet $-a^2-a+2$ på plads 3,3 vil vi gerne ændre til 1 ved en rækkeoperation.
Men det forudsætter, at tallet ikke er 0!

> $solve\left(-a^2-a+2=0, a\right)$

$$-2, 1 \tag{1.3}$$

NB: Matrixelementet kan udtrækkes som $T6[3, 3]$:

> $solve(T6[3, 3]=0, a)$

$$-2, 1 \tag{1.4}$$

Vi ser altså, at vi ikke kan komme videre umiddelbart med $a=-2$.

Vi har bagefter 2 specieltilfælde at undersøge: $a = 1$ og $a = -2$.

Antag: $a \neq 1$ og $a \neq -2$:

> $T7 := RowOperation\left(T6, 3, \dfrac{1}{2 - a^2 - a}\right);$
> $T7 := simplify(T7);$
> $T8 := RowOperation(T7, [2, 3], 1);$
> $T9 := RowOperation(T8, [1, 3], -1 - a);$
> $T9 := simplify(T9)$

$$T7 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 1 & \dfrac{1-a}{-a^2 - a + 2} \end{bmatrix}$$

$$T7 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 1 & \dfrac{1}{a+2} \end{bmatrix}$$

$$T8 := \begin{bmatrix} 1 & 0 & a+1 & 1 \\ 0 & 1 & 0 & \dfrac{1}{a+2} \\ 0 & 0 & 1 & \dfrac{1}{a+2} \end{bmatrix}$$

$$T9 := \begin{bmatrix} 1 & 0 & 0 & 1 + \dfrac{-1-a}{a+2} \\ 0 & 1 & 0 & \dfrac{1}{a+2} \\ 0 & 0 & 1 & \dfrac{1}{a+2} \end{bmatrix}$$

$$T9 := \begin{bmatrix} 1 & 0 & 0 & \dfrac{1}{a+2} \\ 0 & 1 & 0 & \dfrac{1}{a+2} \\ 0 & 0 & 1 & \dfrac{1}{a+2} \end{bmatrix} \qquad \textbf{(1.5)}$$

Delkonklusion:

Hvis $a \in \mathbb{R} \setminus \{-2, 1\}$:

Én løsning, nemlig $\begin{bmatrix} \dfrac{1}{a+2} \\ \dfrac{1}{a+2} \\ \dfrac{1}{a+2} \end{bmatrix}$

Antag: $a = 1$:

> $subs(a = 1, T)$

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 1 \end{bmatrix} \quad \textbf{(1.6)}$$

Nu indgår ingen parameter a, så er det ufarligt at anvende "ReducedRowOperationForm":

> $ReducedRowEchelonForm(subs(a = 1, T))$

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad \textbf{(1.7)}$$

Delkonklusion:

Hvis $a = 1$:

Uendelig mange løsninger givet ved: $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} + t_1 \cdot \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} + t_2 \cdot \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$ hvor $t_1 \in \mathbb{R}$ og $t_2 \in \mathbb{R}$

Antag: $a = -2$:

> $subs(a = -2, T)$

$$\begin{bmatrix} -2 & 1 & 1 & 1 \\ 1 & -2 & 1 & 1 \\ 1 & 1 & -2 & 1 \end{bmatrix} \quad \textbf{(1.8)}$$

Nu indgår ingen parameter a, så er det ufarligt at anvende "ReducedRowOperationForm":

> $ReducedRowEchelonForm(subs(a = -2, T))$

$$\begin{bmatrix} 1 & 0 & -1 & 0 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(1.9)}$$

Delkonklusion:

Hvis $a = -2$:

Ingen løsning.

Samlet konklusion:

Den fuldstændige løsning til ligningssystemet er:

Hvis $a \in \mathbb{R} \setminus \{-2, 1\}$:

Én løsning, nemlig $\begin{bmatrix} \frac{1}{a+2} \\ \frac{1}{a+2} \\ \frac{1}{a+2} \end{bmatrix}$

Hvis $a = -2$:
Ingen løsning.

Hvis $a = 1$:

Uendelig mange løsninger givet ved: $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} + t_1 \cdot \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} + t_2 \cdot \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$ hvor $t_1 \in \mathbb{R}$ og $t_2 \in \mathbb{R}$

Metode 2: Når vi har lært om begrebet *Determinant*, så løses opgaven lettere!

> *Determinant*(A)

$$a^3 - 3a + 2 \quad (2.1)$$

> *solve*$(Determinant(A) = 0, a)$

$$-2, 1, 1 \quad (2.2)$$

Det betyder, at rangen af A er 3, når $\boldsymbol{a} \in \mathbb{R} \setminus \{-2, 1\}$.
Og at rangen er < 3, når $a = -2$ eller $a = 1$.

Den videre løsning må så opdeles i 3 tilfælde!

Antag, at $\boldsymbol{a} \in \mathbb{R} \setminus \{-2, 1\}$.
"LinearSolve" kan anvendes uden problemer, da matricen A er regulær (har determinant $\neq 0$).

> *LinearSolve*(A, b)

$$\begin{bmatrix} \frac{1}{a+2} \\ \frac{1}{a+2} \\ \frac{1}{a+2} \end{bmatrix} \quad (2.3)$$

Dvs. netop én løsning.

Antag nu, at $\boldsymbol{a} = -2$.

Så kan systemet igen løses let, idet vi fortæller Maple, at $a = -2$:

> *LinearSolve*$(subs(a = -2, A), b)$

```
Error, (in LinearAlgebra:-BackwardSubstitute) inconsistent system
```

Dvs. ingen løsning!

Antag nu, at $\boldsymbol{a} = 1$.

Så kan systemet igen løses let, idet vi fortæller Maple, at $a = 1$:

> *LinearSolve*$(subs(a = 1, A), b)$

$$\begin{bmatrix} 1 - \_t1_{2,1} - \_t1_{1,1} \\ \_t1_{2,1} \\ \_t1_{1,1} \end{bmatrix} \quad (2.4)$$

Dvs. uendelig mange løsninger, givet ved 2 frie parametre.

Konklusion:
Den fuldstændige løsning til ligningssystemet er:

Hvis $a \in \mathbb{R} \setminus \{-2, 1\}$:

Én løsning, nemlig $\begin{bmatrix} \frac{1}{a+2} \\ \frac{1}{a+2} \\ \frac{1}{a+2} \end{bmatrix}$

Hvis $a = -2$:

Ingen løsning.

Hvis $a = 1$:

Uendelig mange løsninger givet ved: $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} + t_1 \cdot \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} + t_2 \cdot \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$ hvor $t_1 \in \mathbb{R}$ og $t_2 \in \mathbb{R}$

Lad os undersøge rangen:

NB: Maple skelner ikke mellem værdierne af a, som er specielle! Maple påstår her, at rangen altid er 3.

```
> Rank(A);
  Rank(T)
```

$$3$$
$$3 \tag{2.5}$$

```
> Rank(subs(a =-2, A));
  Rank(subs(a =-2, T))
```

$$2$$
$$3 \tag{2.6}$$

```
> Rank(subs(a = 1, A));
  Rank(subs(a = 1, T))
```

$$1$$
$$1 \tag{2.7}$$

Der er altså løsninger, når blot $a \neq -2$, idet $rang(A) = rang(T)$.

Antal variabel $n = 3$.
Antal frie variable = $n - rang(A)$.

Idet generelle tilfælde er $n - rang(A) = 3 - 3 = 0$, så der er kun den ene løsning.
Når $a = 1$ er $n - rang(A) = 3 - 1 = 2$. Løsningen er således 2-dimensionel.